

ORF 522
Linear Optimization

Lecture 20

Applications in Mechanical Engineering

Trajectory Optimization

Moonshot—The Basics

Position in Euclidean space is a function of time:

$$\mathbf{x}(t).$$

Velocity is the derivative of position:

$$\mathbf{v}(t) = \frac{d\mathbf{x}}{dt}(t).$$

Acceleration is the derivative of velocity:

$$\mathbf{a}(t) = \frac{d\mathbf{v}}{dt}(t).$$

Fixing some convenient coordinate system, all three quantities are vectors:

$$\mathbf{x}(t) = (x_1(t), x_2(t), x_3(t))$$

$$\mathbf{v}(t) = (v_1(t), v_2(t), v_3(t))$$

$$\mathbf{a}(t) = (a_1(t), a_2(t), a_3(t)).$$

Newtonian Mechanics

Newton's Second Law of Motion: in a noninertial frame of reference, the total force vector \mathbf{F} on a body of mass m produces an acceleration given by

$$\mathbf{F} = m\mathbf{a}.$$

The forces on a rocket to the moon are:

1. The gravitational force of attraction to the Earth.
2. The gravitational force of attraction to the Moon.
3. The force θ produced by the rocket's engine.

Newton's Law of Gravitation: two masses m_a and m_b experience an attractive force whose magnitude is given by

$$F = G \frac{m_a m_b}{r^2},$$

where G is a universal constant (called the **Gravitation constant**) and r denotes the distance between the two masses.

The force vector on m_a is therefore:

$$\mathbf{F} = G \frac{m_a m_b}{\|\mathbf{x}_b - \mathbf{x}_a\|^3} (\mathbf{x}_b - \mathbf{x}_a)$$

A Nonconvex Optimization Problem

minimize:

$$cT + \int_0^T \|\theta(t)\|^2 dt$$

subject to:

initial conditions:

$$\mathbf{x}(0) = \text{given Earthbound values}$$

$$\mathbf{v}(0) = \text{given Moonbound values}$$

final conditions:

$$\mathbf{x}(T) = \text{given Moonbound values}$$

$$\mathbf{v}(T) = \text{given Moonbound values}$$

force balance at all times:

$$\begin{aligned} \mathbf{a}(t) = & G \frac{m_e}{\|\mathbf{x}_e - \mathbf{x}(t)\|^3} (\mathbf{x}_e - \mathbf{x}(t)) \\ & + G \frac{m_m}{\|\mathbf{x}_m - \mathbf{x}(t)\|^3} (\mathbf{x}_m - \mathbf{x}(t)) \\ & + \theta(t)/m_r \end{aligned}$$

where:

m_e = mass of the Earth

m_m = mass of the Moon

m_r = mass of the Rocket

x_e = position of the Earth

x_m = position of the Moon

The ampl Model

```

param pi := 4*atan(1);

param n;          # number of intervals in which time is discretized

set D := {1..2};
set Nx := {0..n};          # discrete times for position
set Nv := {0.5..n-0.5 by 1}; # discrete times for velocity
set Na := {1..n-1};        # discrete times for acceleration

param G;
param m_earth;
param r_earth;
param x_earth{D};
param m_moon;
param r_moon;
param x_moon{D};

param x0{D};          # initial position
param xn{D};          # final position
param v0{D};          # initial velocity
param vn{D};          # final velocity
param alpha0;
param alphan;

var T >= 0;          # total time
var x {D, Nx};      # position
var v {d in D, i in Nv} = n*(x[d,i+0.5] - x[d,i-0.5])/T; # velocity
var a {d in D, i in Na} = n*(v[d,i+0.5] - v[d,i-0.5])/T; # acceleration
var theta {D, Na};  # thrust

minimize totalcost: T + (sum {d in D, i in Na} theta[d,i]^2)*(T/n);

subject to init_pos {d in D}: x[d,0] = x0[d];
subject to finl_pos {d in D}: x[d,n] = xn[d];

subject to init_vel {d in D}: v[d, 0.5] = v0[d];
subject to finl_vel {d in D}: v[d, n-0.5] = vn[d];

subject to force_bal {d in D, i in Na}:
    a[d,i] = - G*m_earth*(x[d,i]-x_earth[d])

```

```

      /(sum {dd in D} (x[dd,i]-x_earth[dd])^2)^(3/2)
- G*m_moon*(x[d,i]-x_moon[d])
      /(sum {dd in D} (x[dd,i]-x_moon[dd])^2)^(3/2)
+ theta[d,i];

```

```
data;
```

```

let n := 100;
let T := 100;
let G := 1;
let m_earth := 10; let r_earth := 2; let x_earth[1] := 0; let x_earth[2] := 0;
let m_moon := 1; let r_moon := 1; let x_moon[1] := 20; let x_moon[2] := 0;
let alpha0 := 1.5*pi;
let alphan := pi/2;
let x0[1] := x_earth[1]+r_earth*cos(alpha0);
let x0[2] := x_earth[2]+r_earth*sin(alpha0);
let xn[1] := x_moon[1] +r_moon *cos(alphan);
let xn[2] := x_moon[2] +r_moon *sin(alphan);
let v0[1] := 0.5*cos(alpha0); let v0[2] := 0.5*sin(alpha0);
let vn[1] := -0.5*cos(alphan); let vn[2] := -0.5*sin(alphan);

let {d in D, i in Nx} x[d,i] := (1-i/n)*x0[d] + (i/n)*xn[d];

solve;

printf {i in Nx}: "%10.5f %10.5f \n", x[1,i], x[2,i] > moon.out;

printf {i in Na}: "%10.5f %10.5f \n",
      i*T/n, theta[1,i]^2 + theta[2,i]^2 > fuel.out;

```

The Iteration Log

LOQO 4.01: verbose=2


timing=1

variables: non-neg 1, free 396, bdd 0, total 397
 constraints: eq 202, ineq 0, ranged 0, total 202
 nonzeros: A 1192, Q 2167


Iter	Primal		Dual		Sig Fig	Status	P	M
	Obj Value	Infeas	Obj Value	Infeas				
1	1.0000000e+02	2.84e+00	0.0000000e+00	9.31e+01				
nonzeros: L 3158, arith_ops 26967								
2	9.4224546e+01	3.26e+00	4.0181242e+01	4.72e+00				
3	9.1386817e+01	3.79e+00	7.6313775e+01	6.76e+00	1			
4	9.0660706e+01	3.79e+00	1.3442419e+02	6.97e+00				
5	8.7823928e+01	3.59e+00	4.3655699e+02	4.90e+00				
6	8.6887641e+01	3.07e+00	6.4286207e+02	2.06e+00				
7	8.8524045e+01	2.62e+00	7.2775637e+02	1.27e+00				
8	9.1716411e+01	2.49e+00	7.4408150e+02	1.13e+00				
9	1.0651875e+02	2.38e+00	7.1956806e+02	1.05e+00				
10	1.7363570e+02	1.33e+00	7.3726730e+02	5.70e-01				
11	1.7829697e+02	1.07e+00	7.0675762e+02	4.05e-01			-7	
.								
.								
.								
20	1.1792173e+02	8.61e-03	1.0977773e+02	3.52e-02	1		-10	
21	1.0335840e+02	8.15e-03	8.1588727e+01	2.96e-02	1		-10	
22	8.3891260e+01	1.40e-02	6.3583741e+01	2.59e-02	1		-15	
23	5.2816548e+01	4.54e-02	6.4195010e+01	2.89e-02	1		-15	
24	3.7203453e+01	3.84e-02	7.0793688e+01	2.28e-02			-16	
25	2.5797250e+01	3.44e-02	7.7649750e+01	3.49e-02			-18	
26	2.0629033e+01	2.23e-02	6.5968239e+01	3.86e-02			-4	
27	1.9998731e+01	6.90e-03	3.6427396e+01	2.42e-02				
28	2.0892127e+01	1.16e-03	2.3524441e+01	4.59e-03	1			
29	2.1044356e+01	1.28e-04	2.1304027e+01	5.36e-04	2			
30	2.1064869e+01	7.49e-06	2.1079641e+01	3.19e-05	3			
31	2.1066122e+01	3.79e-07	2.1066915e+01	1.67e-06	4	PF		
32	2.1066186e+01	1.90e-08	2.1066241e+01	1.69e-07	6	PF	DF	
33	2.1066189e+01	9.52e-10	2.1066198e+01	8.68e-08	6	PF	DF	
34	2.1066189e+01	4.77e-11	2.1066190e+01	1.87e-08	8	PF	DF	-9

 OPTIMAL SOLUTION FOUND

The Optimal Trajectory



Fuel Usage



Comparison Shopping

Solver	Time (secs)		
	old	current	
loqo	4.53	17.5	
minos	19.98	21.7	crash
snopt	65.37	87.9	
lancelot	9.14	stuck	20.1

Possible Enhancements

Account for the fact that the mass of the rocket decreases as the fuel is used up.

Put an upper bound on the amount of fuel used.

Provide “real” data (in 3-D).

Account for the motion of the moon.